

# Sinusoidal bearing in 1D: analytical solution

Analytical solution of Reynolds equation with mass-conserving cavitation model (JFO)

Supplementary material to the paper:

J. Lengiewicz, M. Wichrowski and S. Stupkiewicz

*Mixed formulation and finite element treatment of the mass-conserving cavitation model*

*Tribology International*, vol. 72, pp. 143-155 (2014)

<http://dx.doi.org/10.1016/j.triboint.2013.12.012>

Corresponding author:

Stanislaw Stupkiewicz

Institute of Fundamental Technological Research (IPPT PAN), Warsaw, Poland

[ssupkie@ippt.pan.pl](mailto:ssupkie@ippt.pan.pl)

<http://www.ippt.pan.pl/~sstupkie>

The file can be downloaded from:

<http://www.ippt.pan.pl/~sstupkie/files/bearing.html>

version 1.3 (2014-02-01)

---

## Brief description of the problem

This *Mathematica* notebook presents an analytical solution of a 1D hydrodynamic lubrication problem with cavitation for a sine-shaped lubricant film thickness. The mass-conserving JFO cavitation model is used. The problem is defined in our paper, where all the details can be found.

The solution consists of two full-film regions separated by a cavitation region. The one-dimensional steady-state Reynolds equation holds in the full-film regions

$$(1) \quad \frac{d}{dx} \left( u h - \frac{h^3}{12 \eta} \frac{dp}{dx} \right) = 0$$

where

$$(2) \quad h(x) = \frac{1}{2} (h_1 + h_2) - \frac{1}{2} (h_1 - h_2) \cos\left(\frac{\pi x}{L}\right), \quad -L \leq x \leq L$$

Equation (1) is integrated to yield

$$(3) \quad \frac{dp}{dx} = 12 \eta u \frac{h-h^*}{h^3}$$

where  $h^*$  is an unknown integration constant.

The solution in the cavitated region is governed by the following mass-balance equation

$$(4) \quad \frac{d}{dx} ((1 - \lambda) u h) = 0$$

which after integration gives an algebraic equation for the unknown void fraction  $\lambda$

$$(5) \quad (1 - \lambda) u h = u h^*$$

Here, the unknown integration constant  $h^*$  is equal to that in Eq. (3) so that the mass flux is conserved.

Equation (3) is integrated symbolically using *Mathematica*. The position of the film rupture boundary is then obtained numerically by requiring that the pressure and its derivative are equal to zero at this boundary. Finally, the position of the reformation boundary is obtained numerically by requiring that the pressure is equal to zero at this boundary and that the mass flux is preserved in the full film region. The void fraction  $\lambda$  is obtained from the mass-conservation equation (5) in the cavitated region.

Note that direct integration of the Reynolds equation leads to a non-physical discontinuous solution due to the trigonometric functions involved. For this reason special parameterization by  $p_0$  (pressure at  $x = 0$ ) and  $h^*$  (integration constant) is adopted, and appropriate limits for  $x \rightarrow -L$  and  $x \rightarrow L$  are exploited.

**The notebook comes as it is.** It gives the analytical solution for the input parameters used in the original paper. The solution procedure has not been tested for other input parameters. But it should work also for a modified input, possibly after the initial guess for the FindRoot[ ] is also modified.

**Feel free to use and modify this notebook for your work provided the source, i.e. our paper, is adequately cited.**

## Symbolic integration of Reynolds equation

### ■ Clear variables

```
In[1]:= Clear[h, h1, h2, L, ηU, x, p]
```

### ■ Set the numerical values of the input parameters of the problem

Input parameters:

h1p - maximum film thickness

h2p - minimum film thickness

Lp - half-length of the domain

ηUp - product of viscosity  $\eta$  and entrainment speed U

pBC - prescribed pressure at  $x = -L$  and  $x = L$  (boundary conditions)

[Attention! There is a misprint in the paper: the total length of the bearing should be  $l = 125$  mm and not  $l = 12.5$  mm.]

```
In[2]:= h1p = 0.025;
h2p = 0.015;
Lp = 125. / 2;
ηUp = 0.015 * 10-6 * (4000 / 2);
pBC = 1;
```

### ■ Set the numerical values for the initial guess for the root finding function

Rupture boundary:

```
In[7]:= xcini = 25.;
p0cini = 3.;
```

Reformation boundary:


```
In[9]:= xrini = 55.;
p0rini = -2.;
```

### ■ Define film thickness $h(x)$ for $-L \leq x \leq L$

```
In[11]:= h = (h1 + h2) / 2 - (h1 - h2) / 2 Cos[π x / L]
```

```
Out[11]=  $\frac{h_1 + h_2}{2} - \frac{1}{2} (h_1 - h_2) \cos\left[\frac{\pi x}{L}\right]$ 
```

```
In[12]:= Block[ {h1 = h1p, h2 = h2p, L = Lp},
  Plot[ h, {x, -L, L}, PlotRange -> {0, h1}, AxesLabel -> {"x", "h(x)} ] ]
```



### ■ Integrate Reynolds equation in the full-film region

$p_0$  - unknown pressure at  $x = 0$

$h_s$  - unknown integration constant

$h_1, h_2, L, \eta U$  - parameters of the problem

```
In[13]:= pSolution[x_, h1_, h2_, L_, ηU_, hs_, p0_] =
  p[x] /. DSolve[ {p'[x] == 12 ηU / h^3 (h - hs), p[0] == p0}, p[x], x][[1]] // Simplify
```

Out[13]=

$$6 \left( h_1^2 (4 h_2 - 3 h_s) + 2 h_1 h_2 (2 h_2 - h_s) - 3 h_2^2 h_s \right)$$

$$\begin{aligned} & L \eta U \operatorname{ArcTan} \left[ \frac{\sqrt{h_1} \operatorname{Tan} \left[ \frac{\pi x}{2L} \right]}{\sqrt{h_2}} \right] \left( h_1 + h_2 + (-h_1 + h_2) \operatorname{Cos} \left[ \frac{\pi x}{L} \right] \right)^2 + \\ & \sqrt{h_1} \sqrt{h_2} \left( 3 h_1^4 h_2^2 p_0 \pi + 2 h_1^3 h_2^3 p_0 \pi + 3 h_1^2 h_2^4 p_0 \pi - 4 h_1^2 h_2^2 (h_1^2 - h_2^2) p_0 \pi \operatorname{Cos} \left[ \frac{\pi x}{L} \right] + \right. \\ & h_1^2 (h_1 - h_2)^2 h_2^2 p_0 \pi \operatorname{Cos} \left[ \frac{2 \pi x}{L} \right] + 24 h_1^3 h_2 L \eta U \operatorname{Sin} \left[ \frac{\pi x}{L} \right] - \\ & 24 h_1 h_2^3 L \eta U \operatorname{Sin} \left[ \frac{\pi x}{L} \right] - 18 h_1^3 h_s L \eta U \operatorname{Sin} \left[ \frac{\pi x}{L} \right] - 42 h_1^2 h_2 h_s L \eta U \operatorname{Sin} \left[ \frac{\pi x}{L} \right] + \\ & 42 h_1 h_2^2 h_s L \eta U \operatorname{Sin} \left[ \frac{\pi x}{L} \right] + 18 h_2^3 h_s L \eta U \operatorname{Sin} \left[ \frac{\pi x}{L} \right] - 12 h_1^3 h_2 L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] + \\ & 24 h_1^2 h_2^2 L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] - 12 h_1 h_2^3 L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] + 9 h_1^3 h_s L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] - \\ & \left. 9 h_1^2 h_2 h_s L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] - 9 h_1 h_2^2 h_s L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] + 9 h_2^3 h_s L \eta U \operatorname{Sin} \left[ \frac{2 \pi x}{L} \right] \right) \Bigg/ \\ & \left( 2 h_1^{5/2} h_2^{5/2} \pi \left( h_1 + h_2 + (-h_1 + h_2) \operatorname{Cos} \left[ \frac{\pi x}{L} \right] \right)^2 \right) \end{aligned}$$

### ■ Determine pressure at $x = -L$ and $x = L$ as a function of $p_0$ and $h_s$

```
In[14]:= p1 = Limit[ pSolution[x, h1, h2, L, ηU, hs, p0], x -> -L, Direction -> -1 ] //
  Simplify[#, L > 0 && h1 > 0 && h2 > 0 && ηU > 0] &
```

Out[14]=

$$\frac{1}{2 h_1^3 h_2^2} \left( 2 h_1^3 h_2^2 p_0 - 3 \sqrt{\frac{h_1}{h_2}} \left( h_1^2 (4 h_2 - 3 h_s) + 2 h_1 h_2 (2 h_2 - h_s) - 3 h_2^2 h_s \right) L \eta U \right)$$

```
In[15]:= p2 = Limit[ pSolution[x, h1, h2, L, ηU, hs, p0], x → L, Direction → 1 ] //
Simplify[#, L > 0 && h1 > 0 && h2 > 0 && μU > 0 ] &
```

$$\text{Out[15]} = \frac{1}{2 h_1^3 h_2^2} \left( 2 h_1^3 h_2^2 p_0 + 3 \sqrt{\frac{h_1}{h_2}} \left( h_1^2 (4 h_2 - 3 h_s) + 2 h_1 h_2 (2 h_2 - h_s) - 3 h_2^2 h_s \right) L \eta U \right)$$

■ Compute  $h_s$  corresponding to  $p_1=p_{BC}$  (denoted by  $h_{s1}$ ) and to  $p_2=p_{BC}$  (denoted by  $h_{s2}$ )

```
In[16]:= hs1 = hs /. Solve[ p1 == pBC, hs ] [[1]]
```

$$\text{Out[16]} = \frac{2 h_1 h_2 \left( h_1^2 h_2 - h_1^2 h_2 p_0 + 6 h_1 \sqrt{\frac{h_1}{h_2}} L \eta U + 6 \sqrt{\frac{h_1}{h_2}} h_2 L \eta U \right)}{3 \sqrt{\frac{h_1}{h_2}} (3 h_1^2 + 2 h_1 h_2 + 3 h_2^2) L \eta U}$$

```
In[17]:= hs2 = hs /. Solve[ p2 == pBC, hs ] [[1]]
```

$$\text{Out[17]} = \frac{2 h_1 h_2 \left( -h_1^2 h_2 + h_1^2 h_2 p_0 + 6 h_1 \sqrt{\frac{h_1}{h_2}} L \eta U + 6 \sqrt{\frac{h_1}{h_2}} h_2 L \eta U \right)}{3 \sqrt{\frac{h_1}{h_2}} (3 h_1^2 + 2 h_1 h_2 + 3 h_2^2) L \eta U}$$

■ Pressure gradient

```
In[18]:= DpSolution[x_, h1_, h2_, L_, ηU_, hs_, p0_] =
D[pSolution[x, h1, h2, L, ηU, hs, p0], x] // Simplify
```

$$\text{Out[18]} = - \frac{48 \eta U (-h_1 - h_2 + 2 h_s + (h_1 - h_2) \cos\left[\frac{\pi x}{L}\right])}{(h_1 + h_2 + (-h_1 + h_2) \cos\left[\frac{\pi x}{L}\right])^3}$$

■ Find numerically the position  $x_c$  of rupture boundary such that the pressure and its gradient at  $x_c$  are equal to 0

The corresponding pressure  $p_0$  is also computed, and subsequently the integration constant  $h_s$ .

```
In[19]:= Block[ {h1 = h1p, h2 = h2p, L = Lp, ηU = ηUp},
{xc, p0c} = {x, p0} /. FindRoot[
{pSolution[x, h1, h2, L, ηU, hs1, p0], DpSolution[x, h1, h2, L, ηU, hs1, p0]},
{{x, xcini, xcini + 1}, {p0, p0cini, p0cini + 0.1}}
];
hsc = hs1 /. p0 → p0c;
{xc, p0c, hsc}
]
```

```
Out[19]= {20.8491, 3.31208, 0.0175034}
```

■ Find numerically the position  $x_r$  of the reformation boundary such that the pressure is equal to 0

The second equation ( $h_2=h_s$ ) implies that the flux is identical in both full-film regions.

```
In[20]:= Block[ {h1 = h1p, h2 = h2p, L = Lp, ηU = ηUp},
{xr, p0r} = {x, p0} /. FindRoot[
{pSolution[x, h1, h2, L, ηU, hs2, p0], hs2 - hsc},
{{x, xrini, xrini + 1}, {p0, p0rini, p0rini + 0.1}}
];
{xr, p0r}
]
```

```
Out[20]= {56.7045, -1.31208}
```

## ■ Construct the solution

The void fraction  $\lambda$  in the cavitation zone is computed from the mass balance equation:  $(1 - \lambda)h U = h^* U$

```
In[21]:= Block[ {h1 = h1p, h2 = h2p, L = Lp, ηU = ηUp},
  pAnalytical[x_] = Piecewise[
    {{pSolution[x, h1, h2, L, ηU, hsc, p0c], x ≤ xc},
     {0, xc < x ≤ xr},
     {pSolution[x, h1, h2, L, ηU, hsc, p0r], xr < x}}];
  λAnalytical[x_] = Piecewise[
    {{0., x ≤ xc},
     {λ /. Solve[(1 - λ) h == hsc, λ][[1]], xc < x ≤ xr},
     {0., xr < x}}];
]
```

## ■ Plot the solution

```
In[22]:= Plot[pAnalytical[x], {x, -Lp, Lp}, Axes → False,
  Frame → True, FrameLabel → {"x", "p"}, PlotRange → {-0.4, 7.4}]
Plot[λAnalytical[x], {x, -Lp, Lp}, Axes → False, Frame → True,
  FrameLabel → {"x", "λ"}, PlotRange → {-0.02, 0.32}]
```

